
Алгоритми пошуку в олімпіадних задачах

Дану тему слід розглядати в двох аспектах. По-перше, при рішенні різних олімпіадних задач алгоритми пошуку потрібно використовувати в якості технічних елементів при реалізації власного алгоритмові рішення тої чи іншої задачі. По-друге, задача сама по собі може потребувати побудови оптимального в якості визначених вимог алгоритмові пошуку. Підхід у кожному з двох згаданих випадків повинен бути різним.

Алгоритми пошуку в неупорядкованих одномірних масивах.

Розглянемо спочатку нюанси реалізації “технічних” задач пошуку, що зустрічаються в різних алгоритмах. Почнемо з, здавалося б, тривіальної задачі пошуку елемента в неупорядкованому масиві. В усіх прикладах, що відносяться до пошуку в одномірному масиві будемо використовувати наступну змінну a:

a:array[0..N] of <скалярний тип>;

при цьому власне елементи масиву, що ми будемо розглядати, пронумеровані від 1 до N, а нульовий елемент будемо використовувати як допоміжний у разі потреби. Конкретний же тип елемента в більшості описуваних алгоритмів не важливий, він може бути як будь-яким числовим, так і символьним або навіть строковим. Алгоритм пошуку в масиві елемента, значення якого дорівнює K, може виглядати так:

i:=0;

repeat

  i:=i+1

until (i=N) or (a[i]=K);

if a[i]=K then write(i)

          else write(0)

{наступне невірно (!!!):

 if i=N then write(0)

        else write(i) }

При відсутності в масиві елемента із шуканим значенням K друкується значення нульового індексу. Виявляється, що і таку досить просту програму можна спростити, заодно продемонструвавши так називаний “бар'єрний” метод, що часто застосовується в програмуванні в тому числі і олімпіадних задач. У даному випадку він полягає в тому, що ми можемо занести в додатковий елемент масиву (наприклад, нульовий) шукане значення K, позбувшись тим самим в умові закінчення циклу від перевірки на вихід за кордон масиву:

a[0]:=K;

i:=N;

while (a[i]<>K) do

  i:=і-1;

write(i)

Ця програма не тільки простіше і ефективніше. У ній практично неможливо зробити помилку.

Інший корисний прийом можна показати на задачі пошуку максимального і мінімального значення в масиві. Він полягає в тому, що при будь-якому пошуку в масиві шукати випливає не значення шуканого елемента, максимуму, мінімуму і т.п., а його індекс. Тоді вирішуючи одну задачу, ми по суті справи вирішуємо відразу дві: визначаємо не тільки наявність шуканого елемента, значення максимуму або мінімуму, але і їх розташування в масиві. Програма при цьому не тільки не ускладнюється, але найчастіше стає навіть коротше:

imax:=1;

imin:=1;

for i:=2 to N do
  if a[i]<a[imin] then imin:=i else
  if a[i]>a[imax] then imax:=i

Помітимо, що використання як початкових значення для мінімуму і максимуму не значення першого елемента масиву, а максимальне і мінімальне значення в типі елементів, може привести до помилки. Так програма не знаходить значення максимуму для убуваючого масиву цілих чисел (!!!):

max:=-MaxInt-1;{це мінімальне число типу integer}

min:=MaxInt;{це максимальне число типу integer}

for i:=1 to N do
  if a[i]<min then min:=a[i] else
  if a[i]>max then max:=i

Здавалося б на цьому розгляд алгоритмів пошуку в неупорядкованому масиві можна завершити. Однак саме з одночасного пошуку мінімуму і максимуму можна почати клас алгоритмів пошуку, більш ефективних, ніж приведені вище стандартні алгоритми. Причому саме при рішенні олімпіадних задач їх знання може бути потрібним в першу чергу. Отже, у приведеному вище алгоритмі пошуку максимального і мінімального елемента в масиві в гіршому випадку виконується 2N – 2 порівнянь.Припустимо, що ту ж задачу можна вирішити за 3N/2 – 2 порівняння*. Нехай у нас є парне число елементів. Розіб'ємо їх на пари і у кожній з N/2 пар за одне порівняння визначимо, який елемент більше, а який менше. Тоді максимум можна шукати вже будь-яким способом тільки з найбільших елементів у парах, а мінімум — серед найменших. Загальне число порівнянь при цьому дорівнює N/2 + 2(N/2 – 1) = 3N/2 – 2. Для непарного числа елементів — елемент, що не потрапив у жодну з пар, повинний брати участь як у пошуку максимуму, так і мінімуму.

{виноска*}

Позначення  відповідає округленню до найближчого цілого числа, більшого або рівного вираженню в зазначених дужках, на відміну від цілої частини, де округлення відбувається до найближчого цілого, меншого або рівного розглянутому виразу.

{\виноска}

Ще більшої ефективності можна домогтися при одночасному пошуку максимальній і другого по величині числа. Для цього організуємо пошук максимуму за схемою “тенісного турніру”, а саме: розіб'ємо елементи на пари і визначимо в кожній з пар більший елемент, потім розіб'ємо на пари вже ці елементи і т.д. У “фіналі” і буде визначений максимум. Кількість порівнянь у такому алгоритмі, як і в традиційній схемі, дорівнює N – 1. Однак, максимальний елемент брав участь при цьому в log2N порівняннях, причому одне з них проходило обов'язково з другим по величині елементом. Таким чином, тепер для пошуку цього елемента буде потрібно усього log2N – 1 порівняння (!!!). Спробуйте самостійно побудувати ефективний алгоритм для пошуку вже перших трьох по величині елементів.

У даному контексті можна поставити також задачу пошуку i-го по рахунку елемента, що називається i-ої порядковою статистикою. Крім максимуму і мінімуму, специфічною порядковою статистикою є медіана — елемент із номером (N + 1)/2 для непарних N і два сусідніх елементи для парних. Звичайно, задачу пошуку будь-якої порядкової статистики можна вирішити, попередньо відсортувавши масив. Але, як буде показано нижче, оптимальні по кількості порівнянь універсальні (тобто придатні для довільних масивів) алгоритми сортування виконують порядку Nlog2N порівнянь. А задачу пошуку i-ої порядкової статистики можна вирішити, роблячи O(N) порівнянь. Алгоритм, що гарантує цю оцінку досить складний, він докладно викладений у [1, 2, 3]. Ми ж приведемо схему алгоритму, що у гіршому випадку не є лінійним, однак на практиці працює істотно швидше, отже саме його і потрібно використовувати в практиці реального програмування, у тому числі і олімпіадних задач:

Алгоритм Вибір(A, L, R, і)

{вибирає між L-ым і R-ым елементами масиву A

 i-ий по рахунку в порядку зростання елемент}

1. if L=R then результат – A[L], кінець;

2. Q:=L+random(R-L+1)
 {Q – випадковий елемент між L і R}

3. Переставляємо елементи від L до R у A так, щоб спочатку йшли елементи, менші A[Q], а потім всі інші, перший елемент у другій групі позначимо K.

4. if i((K-L) then Вибір(A, L, K-1, i)
             else Вибір(A, K, R, i-(K-L))

5. кінець.

Оптимальну реалізацію пункту 3 можна запозичити з алгоритму так званого швидкого сортування (див., наприклад, [4]).

Нарешті, розглянемо задачу пошуку в послідовності з N чисел, збереження якої не потрібно взагалі, отже її довжина може бути дуже велика. Нехай відомо, що в цій послідовності зустрічаються по одному разу всі числа від 0 до N, крім одного. Це пропущене число і потрібно знайти. Отже можливе рішення такої задачі:

S:=0;

for i:=1 to N do

begin

  read(a); S:=S+a

end;

writeln(N*(N + 1) div 2 – S)

Дану програму легко переписати так, щоб вона працювала і для значень N, що перевищують максимальне ціле число. Для цього варто використовувати змінні типу extended, а цикл for замінити на while. Використовуючи аналогічний прийом спробуйте вирішити наступну задачу. Нехай N — кількість чисел, непарно і відомо, що серед чисел, що вводяться, кожне має рівно одне, рівне собі, а одне число - немає. Знайдіть це число. (Вказівка. У даному випадку можна скористатися властивістю логічної функції “ що виключає або”, яка позначається в Паскалі як xor: a xor b xor b = a.)

Про інші алгоритми пошуку в одномірних масивах можна прочитати, наприклад, у [2, 3, 5, 6]. Особливий інтерес серед них представляє задача міжнародної олімпіади по інформатиці 2000 року, присвячена перебуванню медіани в масиві, причому не шляхом порівняння елементів між собою, а за допомогою операції визначення середнього елемента серед трьох довільних елементів масиву, виконуваної бібліотечним модулем, наданим учасникам олімпіади.

Пошук в упорядкованих масивах

Під упорядкованими надалі будуть розумітися неубутні масиви, якщо не обговорене інше. Тобто, a[1] ( a[2] ( … ( a[N]...

Спочатку приведемо приклад реалізації широко відомого алгоритму подвійного (бінарного) пошуку елемента, рівного K, в уже впорядкованому масиві. Виявляється, достроковий вихід з циклу у випадку перебування елемента виграшу по швидкості практично не дає, а зайві перевірки роблять програму більш громіздкою. Тому рекомендується робити пошук, поки діапазон розглянутих елементів складається більш, ніж з одного елемента.

L:=1; R:=N+1;

while L<R do
begin

  m:=(L+R)div 2;

  if a[m]<K then L:=m+1

            else R:=m

end;

if a[m]=K then write(m) else write(0)

На півфіналі чемпіонату світу з програмування серед студентських команд вузів, що проходили в м.Санкт-Петербург у 2000 році пропонувалася наступна зворотна бінарному пошукові задача (див. сайт neerc.ifmo.ru на якому можна знайти тести і відповіді до них для зазначеної задачі). Відомо, що алгоритм бінарного пошуку, аналогічний приведеному вище, але який закінчує свою роботу у випадку дострокового виявлення елемента, за Q порівнянь визначив, що шуканим є елемент із номером I. Яка могла бути при цьому розмірність масиву (вказати всі припустимі діапазони значень N). Незважаючи на вдавану складність, при заданих в умові обмеженнях задача вирішувалася шляхом простого перебору різних значень N і звертання з кожним з цих значень до функції бінарного пошуку. Конструктивний же підхід до рішення задачі набагато складніший. Однак і у випадку підбору можна використовувати чимало цікавих фактів. Наприклад, довжина кожного з діапазонів можливих значень N, є ступенем двійки, а кожен наступний діапазон не менше попереднього. Це дозволяє значно скоротити кількість розглянутих значень N. Крім того, максимальне припустиме значення для N легко знайти аналітично.

Розглянемо тепер задачу пошуку в упорядкованому масиві найбільшої "рівнинної" ділянки. Тобто, потрібно знайти таке число p, що в масиві має послідовність з p рівних елементів і немає послідовності з p + 1 рівних за значенням елементів (див., наприклад, [7]). Виявляється, існує алгоритм розв’язку цієї задачі, кількість операцій у якому може виявитися істотно менше, ніж N. Так, якщо ми вже знайшли послідовність з p1 рівних між собою чисел, то іншу послідовність має сенс тепер розглядати тільки якщо її довжина не менш p1 + 1. Тому, якщо a[i] — перший елемент передбачуваної нової підпослідовності, то порівнювати його треба відразу з елементом a[i+p], де p — максимальна величина для вже розглянутих підпослідовностей з рівних елементів. Приведемо фрагмент програми, що вирішує дану задачу. Як результат ми одержуємо довжину максимальної підпослідовності p, номер елемента, з якого ця підпослідовність починається, k і значення елементів у знайденій підпослідовності:

p:=1; k:=1;

i:=1; f:=false;

while i+p<=N do
  if a[i+p]=a[i] then
  begin

    p:=p+1; f:=true
  end

  else if f then 

         begin 

           k:=i; i:=i+p; f:=false 

         end
       else i:=i+1;

writeln(p,’ ’,k, ’ ’,a[k])

У [7] можна знайти ще одну цікаву задачу за назвою “шахрай на посібнику” - пошуку в упорядкованих послідовностях уже практично якої завгодно довжини. Нехай є три упорядкованих за алфавітом списку з прізвищ людей, що одержує допомогу з безробіття в трьох різних місцях. Довжина кожного зі списків може бути як завгодно великою (кожний зі списків можна зберігати в окремому файлі). Відомо, що принаймні одна особа фігурує у всіх трьох списках. Потрібно написати програму пошуку такої особи, порядок кількості операцій у якій не буде більшим, ніж сума довжин усіх трьох списків. Приведемо фрагмент програми, що працює з трьома файлами, звертання до елементів яких (вони можуть бути будь-як типу, у тому числі і string, до елементів якого в Паскалі застосовні операції порівняння, читання і запису) із програми відбувається за допомогою файлових перемінних x, y, z типу text:

readln(x,p); readln(y,q); readln(z,r);

while not((p=q)and(q=r)) do

begin

  if p<q then readln(x,p)

  else if q<r then readln(y,q)

       else if r<p then readln(z,r)

end;

writeln(p);{p=q=r}

Покажіть, що для поставленої задачі читання з файлу завжди буде працювати коректно і на кожнім кроці циклу значення однієї з трьох змінних p, q, r буде змінено. Крім того доведіть, що за допомогою цієї програми завжди буде знайдений саме мінімальний з елементів, що є присутнім у всіх трьох файлах.

Нарешті, розглянемо задачу пошуку елемента, значення якого дорівнює K, у двомірному масиві, упорядкованому по рядках і стовпцям: a[i,j] (a [i,j+1], a[i,j] ( a[i+1,j], 1 ( i < N, 1 ( j < M. Дана задача вирішується за M + N дій, а не за M*N, як у довільному масиві. Пошук варто починати з елемента a[N,1]. Цей елемент самий маленький у своєму рядку і найбільшоий у своєму стовпці (у математиці подібні елементи називають “сідловими крапками”). Тому, якщо він виявиться менше, ніж K, то з розгляду перший стовпець можна виключити, а якщо більше — з розгляду виключається останній рядок і т.д. Отже можлива реалізація цього алгоритму:

i:=N; j:=1;

while (i>0)and(j<M+1)and(a[i,j]<>K) do

  if a[i,j]<K then j:=j+1

  else i:=і-1;

if (i>0)and(j<M+1) then writeln(a[i,j])

Програма могла б бути ще коротше і ефективніше, якби в ній використовувався згаданий вище бар'єрний метод. У даному випадку для організації бар'єра потрібно доповнити масив нульовим рядком і m+1-м стовпцем. В усі створені бар'єрні елементи варто помістити значення K. Тоді умову в циклі можна скоротити до наступних: a[i,j]<>K.

Алгоритми пошуку і задачі на зважування

Поява подібних задач, наприклад задачі визначення фальшивої монети, у контексті розгляду алгоритмів пошуку не випадково. По-перше, пошук і в цьому випадку здійснюється шляхом операцій порівняння, щоправда, уже не тільки одиночних елементів, але і груп елементів між собою. По-друге, як буде показано нижче, на відміну від олімпіад по математиці для молодших школярів, задачі на зважування цілком можна вирішувати конструктивно.

Як приклад розглянемо задачу, що пропонувалася на теоретичному турі однієї з регіональних олімпіад по інформатиці. Нехай у нас є 12 монет, одна з яких фальшива, яка по вазі відрізняється від інших монет, причому невідомо, легше вона чи тяжче. Потрібно за три зважування визначити номер фальшивої монети (спробуйте вирішити цю задачу самостійно і ви переконаєтеся, що це зовсім не просто, а часом узагалі здається неможливим). Введемо наступні позначення. Знаком “+” будемо позначати монети, що під час поточного зважування варто покласти на ваги, причому, якщо монета на вагах уже була, то на ту ж саму чашу, на якій ця монета знаходилася під час свого попереднього зважування. Знаком “-“ будемо позначати монети, які варто перекласти на протилежну чашу ваг, стосовно тієї, на якій вони знаходилися (кожна окремо), помітимо, що якщо монета на вагах ще не була, то знак “-“ до неї застосований бути не може. Нарешті, знаком “0” — монети, що у черговому зважуванні не беруть участі. Тоді, існує 14 різних способів позначення монет для трьох зважувань:

1 2 3 4 5 6 7 8 9 10 11 12 13 14

+ + + + + + + + +  0  0  0  0  0 — перше зважування

+ + + - - - 0 0 0  +  +  +  0  0 — друге зважування

+ - 0 + - 0 + - 0  +  -  0  +  0 — третє зважування

З отриманої таблиці викреслимо 2 стовпці так, щоб у кожному з трьох рядків кількість ненульових елементів виявилося парним (адже ми не можемо під час одного зважування покласти на дві чаші ваг непарне число монет). Це можуть бути, наприклад, стовпці 4 і 14. Тепер будемо зважувати 12 монет так, як це записано в ті що залишилися 12 стовпцях. Тобто, у першому зважуванні будуть брати участь 8 довільних монет, у другому — три монети треба з ваг забрати, дві — перекласти на протилежні стосовно першого зважування чаші ваг і три монети покласти на ваги вперше (на вільні місця так, щоб на кожній з чаш знову виявилося по 4 монети). Відповідно до схеми проведемо і третє зважування, знову розташовуючи на кожній чаші ваг по 4 монети. Результат кожного зважування окремо ніяк не аналізується, а просто записується. При цьому рівновага на вагах завжди кодується нулем, вперше виниклий нерівновагий стан — знаком плюс, якщо при наступному зважуванні ваги відхиляться від рівноваги в ту ж саму сторону, то результат такого зважування також кодується плюсом, а якщо в іншу сторону — то мінусом. Наприклад, записи “=<<” і “=>>” кодуються як “0++”, а запису “<=>” і “>=<” — як “+0-“. Тому що ми не знаємо, легше або тяжче інших монет виявиться фальшива, то нам важливо як змінювався стан ваг від зважування до зважування, а не те яка саме чаша виявлялася тяжче, а яка легше. Тому два, на перший погляд, різних результати трьох зважувань у цьому випадку кодуються однаково. Після подібного запису результатів зважувань фальшива монета уже фактично визначена. Нею виявляється та, котрії відповідає такий же стовпець у таблиці, як і закодований нами результат трьох зважувань. Для першого з прикладів це монета, що брала участь у зважуваннях за схемою, зазначеною в 10-м стовпці таблиці, а для другого — у 8-м. Справді, стан ваг у нашій задачі міняється в залежності від того, де виявляється фальшива монета під час кожного зі зважувань. Тому монета, “поводження” якої погодиться з записаним результатом зважувань, такий результат і визначає.

Аналіз таблиці показує, що цю задачу можна вирішити не тільки для 12, але і для 13 монет. Для цього варто виключити з розгляду будь-якого не утримуючого нулів стовпець, наприклад, все той же четвертий. В іншому всі дії залишаються незмінними. Для довільного числа монет N>2 кількість зважувань визначається по формулі log3(2*N + 1) (за одне зважування задача не розв'язна ні для якої кількості монет!!!), але підхід до рішення задачі при цьому не зміниться.

Спробуйте тепер вирішити задачу, що пропонувалася в 1998 році на вже згадуваному вище півфіналі чемпіонату світу з програмування серед студентських команд вузів. У ній також було потрібно визначити номер фальшивої монети, вага якої відрізнялася від інших. Але всі зважування вже були проведені, а їх результати записані. Число зважувань було вхідним параметром у задачі (воно могло бути і надлишковим у порівнянні з описаним вище оптимальним алгоритмом визначення номера фальшивої монети). При цьому в кожнім зі зважувань могло брати участь будь-яка парна кількість наявних монет (скільки і які саме — відомо). Результати записувалися за допомогою символів “<”, “>” і “=”.

Ще одна задача на зважування розглянута в [8]. У загальному випадку в ній потрібно знайти набір з мінімальної кількості гир такий, що з його допомогою можна зважити будь-який вантаж, що важить ціле число кілограм, у діапазоні від 1 кг до N кг. При необхідності гирі можна розташовувати на обох чашах ваг. Так, для N=40 це гирі 1, 3, 9 і 27 кг.

Пошук підстроки в рядку

Формалізувати цю задачу можна в такий спосіб. Нехай заданий масив s з N елементів (рядок) і масив p з M елементів (підстрока), причому 0<M(N. Потрібно знайти перше безперервне входження p у s. Ця задача на практиці зустрічається дуже часто. Так, у більшості текстових редакторів реалізована операція пошуку за зразком, що практично цілком збігається з описаною задачею. Якщо розмір масиву s — N не перевершує 255, а тип його елементів — char, то в Турбо Паскалі такий пошук можна виконувати за допомогою стандартної функції Pos(p,s). Однак, у загальному випадку її приходиться реалізовувати самостійно. Прямий пошук, заснований на послідовному порівнянні підстроки спочатку з першими M символами рядка, потім із символами з номерами 2 — M+1 і т.д., у гіршому випадку зробить порядку N*M порівнянь. Але для цієї задачі відомий алгоритм Боуера і Мура (див., наприклад, [5]), що для довільних рядків виконує не набагато більш N/M порівнянь. Тобто різниця в обчислювальній складності складає M2 (!!!). Розглянемо останній алгоритм, на прикладі якого також можна показати, що використання невеликої кількості додаткової пам'яті (у даному випадку допоміжного масиву, розмір якого дорівнює розмірові алфавіту рядків) дозволяє істотно прискорити виконання програми.

Перед фактичним пошуком, для всіх символів, що можуть зустрітися в рядку, обчислюється і запам'ятовується в масиві d відстань від самого правого входження цього символу в шукану підстроку до її кінця. Якщо ж якогось символу з алфавіту рядка в підстроке не має, то така відстань вважається рівною довжині підстроки M. Посимвольне  порівняння підстроки з деяким фрагментом рядка починається не з початку, а з кінця шуканої підстроки (зразка). Якщо який-небудь символ зразка не збігається з відповідним символом фрагмента рядка, а х -останній символ фрагмента рядка, то зразок можна зрушити уздовж рядка вправо на d[x] символів. Якщо більшість символів у рядку відмінні від символів подстроки, то зрушення буде відбуватися на M елементів, що і забезпечить приведену вище складність алгоритму. Покажемо роботу алгоритму на прикладі пошуку слова коала в рядку:

кокаколулюбитикоала.

коала

    коала

     коала

          коала

              коала

Тут підкреслені символи, що брали участь у порівняннях. Зрушення визначалися такими значеннями масиву d: d['к']=4, d['л']=1, d['ю']=5. Якби останньої в розглянутому фрагменті рядка виявилася буква а, то величина зрушення була б 2, тому що в зразку є ще одна така буква, що стоїть від кінця на 2 символи, а при її відсутності зрушення було б 5. Приведемо тепер можливу реалізацію описаного алгоритму, для простоти вважаючи, що розмір підстроки не перевершує 255 

const nmax=10000;

var p:string; {підстрока}

  s:array[1..nmax]of char; {рядок}

  d:array[char]of byte; {масив зрушень}

  c:char;

  m,i,j,k:integer;

begin

  …{задання рядка і підстроки}

  m:=length(p);{довжина підстроки}

  for c:=chr(0) to chr(255) do d[c]:=m;

  for j:=1 to m-1 do d[p[j]]:=m-j;

  {масив d визначений}

  i:=m+1;

  repeat {вибір фрагмента в рядку}

    j:=m+1; k:=i;

    repeat {перевірка збігу}

      k:=k-1; j:=j-1

    until (j<1)or(p[j]<>s[k]);

    i:=i+d[s[і-1]];{зрушення}

  until (j<1)or(i>nmax+1);

  if j<1 then write(k+1) else write(0)

end.

Приведений алгоритм не дає виграшу тільки в одному випадку — коли кількість часткових збігів шуканої підстроки з фрагментами тексту досить велика. Це можливо, наприклад, при надзвичайній обмеженості алфавіту, із символів якого складаються рядки. Тоді варто застосовувати алгоритм Кнута-Мориса-Пратта, описаний у [5], або комбінацію з двох алгоритмів.

Розглянуту проблему не слід плутати з такою задачею. Нехай заданий масив s з N елементів і масив p з M елементів, причому 0<M(N. Потрібно з'ясувати, чи можна з першого масиву викреслити деякі члени так, щоб він збігся з другим. Число операцій у даному випадку має порядок N + M.

